Calculation of the asymptotic formula for the sequence A005366

(Vaclav Kotesovec, general formula published in OEIS Apr 01 2021)

In[∘]:=
```
Table[1 + Sum[Product[Binomial[n + d - 1 - k, d] / Binomial[d + k, d] /. d → 8, {k, 0, h}],
    {h, 0, n - 1}], {n, 1, 10}]
```

Out[∘]=
```
{2, 11, 92, 1157, 19142, 403691, 10312304, 311348897, 10826298914, 426196716090}
```

The product can be expressed using the BarnesG function

$$\text{Product}\left[\frac{k!\,(-1+d-h+k+n)!}{(d+k)!\,(-1-h+k+n)!},\ \{k, 0, h\}\right]$$

In[∘]:=
```
Product[1 / Product[j, {j, k + 1, k + d}], {k, 0, h}] // FullSimplify
```

Out[∘]=

$$\frac{\text{BarnesG}[2+d]\,\text{BarnesG}[2+h]\left(\frac{\text{BarnesG}[3+d]}{\text{BarnesG}[2+d]\,\text{Gamma}[2+d]}\right)^{h}}{\text{BarnesG}[2+d+h]\,\text{Gamma}[1+d]}$$

$$\frac{\text{BarnesG}[2+d]\ \text{BarnesG}[2+h]}{\text{BarnesG}[2+d+h]\,\text{Gamma}[1+d]}$$

In[∘]:=
```
Product[Product[j, {j, (-h + k + n), (-1 + d - h + k + n)}], {k, 0, h}]
```

Out[∘]=

$$\frac{1}{\text{BarnesG}[1+n]}\,\text{BarnesG}[1+d+n]\,\text{BarnesG}[1-h+n]^{1-h}\,\text{BarnesG}[1+d-h+n]^{-1+h}$$
$$\text{BarnesG}[2+d-h+n]^{-h}\,\text{Pochhammer}[-h+n, d]\,(\text{BarnesG}[2-h+n]\,\text{Pochhammer}[1-h+n, d])^{h}$$

In[4]:=

$$\frac{\text{BarnesG}[2+d]\ \text{BarnesG}[2+h]}{\text{BarnesG}[2+d+h]\,\text{Gamma}[1+d]} * \frac{1}{\text{BarnesG}[1+n]}\,\text{BarnesG}[1+d+n]\ \text{BarnesG}[1-h+n]^{1-h}$$
$$\text{BarnesG}[1+d-h+n]^{-1+h}\ \text{BarnesG}[1+d-h+n]^{-h}\,\text{Gamma}[1+d-h+n]^{-h}$$
$$\text{Pochhammer}[-h+n, d]\ (\text{BarnesG}[1-h+n]\,\text{Gamma}[1-h+n]\,\text{Pochhammer}[1-h+n, d])^{h}\ //$$
```
FunctionExpand // PowerExpand // Simplify
```

Out[4]=

$$\frac{\text{BarnesG}[2+d]\,\text{BarnesG}[2+h]\,\text{BarnesG}[1+d+n]\,\text{BarnesG}[1-h+n]\,\text{Gamma}[d-h+n]}{\text{BarnesG}[2+d+h]\,\text{BarnesG}[1+n]\,\text{BarnesG}[1+d-h+n]\,\text{Gamma}[1+d]\,\text{Gamma}[-h+n]}$$

Check of first terms for d=8, OK

In[∘]:=

$$\text{Table}\Big[1 + \text{Sum}\Big[\frac{\text{BarnesG}[2+d]\,\text{BarnesG}[2+h]\,\text{BarnesG}[1+d+n]\,\text{BarnesG}[1-h+n]\,\text{Gamma}[d-h+n]}{\text{BarnesG}[2+d+h]\,\text{BarnesG}[1+n]\,\text{BarnesG}[1+d-h+n]\,\text{Gamma}[1+d]\,\text{Gamma}[-h+n]}\ /.\ d \to 8,\ \{h, 0, n-1\}\Big],\ \{n, 1, 10\}\Big]$$

Out[∘]=
```
{2, 11, 92, 1157, 19142, 403691, 10312304, 311348897, 10826298914, 426196716090}
```

The maximum occurs around h = n/2

In[∘]:=
```
With[{d = 8, n = 100}, ListPlot[Table[
   Log@Product[Binomial[n + d - 1 - k, d] / Binomial[d + k, d], {k, 0, h}], {h, 0, n - 1}]]]
```

Out[∘]=



Value at the maximum

In[∘]:=
```
(BarnesG[2 + d] BarnesG[2 + h] BarnesG[1 + d + n] BarnesG[1 - h + n] Gamma[d - h + n]) /
 (BarnesG[2 + d + h] BarnesG[1 + n] BarnesG[1 + d - h + n] Gamma[1 + d] Gamma[-h + n]) /. h → n / 2
```

Out[∘]=

$$\frac{\text{BarnesG}[2+d]\ \text{BarnesG}\left[1+\frac{n}{2}\right]\ \text{BarnesG}\left[2+\frac{n}{2}\right]\ \text{BarnesG}[1+d+n]\ \text{Gamma}\left[d+\frac{n}{2}\right]}{\text{BarnesG}\left[1+d+\frac{n}{2}\right]\ \text{BarnesG}\left[2+d+\frac{n}{2}\right]\ \text{BarnesG}[1+n]\ \text{Gamma}[1+d]\ \text{Gamma}\left[\frac{n}{2}\right]}$$

In[∘]:=
```
Series[(BarnesG[2 + d] BarnesG[1 + n/2] BarnesG[2 + n/2] BarnesG[1 + d + n] Gamma[d + n/2]) /
  (BarnesG[1 + d + n/2] BarnesG[2 + d + n/2] BarnesG[1 + n] Gamma[1 + d] Gamma[n/2]),
  {n, Infinity, 0}] // Normal
```

Out[∘]=

$$\frac{2^{-d}\ e^{\frac{d-2d^2-d^3}{2n}+d\,n\,\text{Log}[2]+\frac{1}{2}\left(2\,d\,\text{Log}[2]+2\,d^2\,\text{Log}[2]-2\,d\,\text{Log}[n]-d^2\,\text{Log}[n]-d\,\text{Log}[2\pi]\right)}\ n^d\ \text{BarnesG}[2+d]}{\text{Gamma}[1+d]}$$

In[∘]:=
```
2^-d e^(d n Log[2] + 1/2 (2 d Log[2] + 2 d^2 Log[2] - 2 d Log[n] - d^2 Log[n] - d Log[2 π])) n^d BarnesG[1 + d] // FullSimplify
```

Out[∘]=

$$2^{d\left(-\frac{1}{2}+d+n\right)}\ n^{-\frac{d^2}{2}}\ \pi^{-d/2}\ \text{BarnesG}[1+d]$$

Contribution of other terms around the maximum

In[∘]:=
```
(BarnesG[2 + d] BarnesG[2 + h] BarnesG[1 + d + n] BarnesG[1 - h + n] Gamma[d - h + n]) /
 (BarnesG[2 + d + h] BarnesG[1 + n] BarnesG[1 + d - h + n] Gamma[1 + d] Gamma[-h + n]) /.
  h → n / 2 - x
```

Out[∘]=

$$\frac{\text{BarnesG}[2+d]\ \text{BarnesG}[1+d+n]\ \text{BarnesG}\left[2+\frac{n}{2}-x\right]\ \text{BarnesG}\left[1+\frac{n}{2}+x\right]\ \text{Gamma}\left[d+\frac{n}{2}+x\right]}{\text{BarnesG}[1+n]\ \text{BarnesG}\left[2+d+\frac{n}{2}-x\right]\ \text{BarnesG}\left[1+d+\frac{n}{2}+x\right]\ \text{Gamma}[1+d]\ \text{Gamma}\left[\frac{n}{2}+x\right]}$$

In[∘]:=
```
% / (BarnesG[2 + d] BarnesG[1 + n/2] BarnesG[2 + n/2] BarnesG[1 + d + n] Gamma[d + n/2] / (BarnesG[1 + d + n/2] BarnesG[2 + d + n/2] BarnesG[1 + n] Gamma[1 + d] Gamma[n/2])) // FullSimplify
```

Out[∘]=

$$\left(\text{BarnesG}\left[1+d+\frac{n}{2}\right]\text{BarnesG}\left[2+d+\frac{n}{2}\right]\text{BarnesG}\left[2+\frac{n}{2}-x\right]\text{BarnesG}\left[1+\frac{n}{2}+x\right]\text{Gamma}\left[\frac{n}{2}\right]\text{Gamma}\left[d+\frac{n}{2}+x\right]\right)\Big/\left(\text{BarnesG}\left[1+\frac{n}{2}\right]\text{BarnesG}\left[2+\frac{n}{2}\right]\text{BarnesG}\left[2+d+\frac{n}{2}-x\right]\text{BarnesG}\left[1+d+\frac{n}{2}+x\right]\text{Gamma}\left[d+\frac{n}{2}\right]\text{Gamma}\left[\frac{n}{2}+x\right]\right)$$

In[∘]:=
```
Series[
  Log[(BarnesG[1 + d + n/2] BarnesG[2 + d + n/2] BarnesG[2 + n/2 - x] BarnesG[1 + n/2 + x] Gamma[n/2]
    Gamma[d + n/2 + x]) / (BarnesG[1 + n/2] BarnesG[2 + n/2] BarnesG[2 + d + n/2 - x]
    BarnesG[1 + d + n/2 + x] Gamma[d + n/2] Gamma[n/2 + x])],
  {n, Infinity, 1}] // Normal // FullSimplify
```

Out[∘]=

$$-\frac{2\,d\,(-2+x)\,x}{n}$$

In[5]:=
```
Assuming[{d > 0, n > 0}, Integrate[Exp[-(2 d (-2 + x) x)/n], {x, -Infinity, Infinity}]]
```

Out[5]= $e^{\frac{2d}{n}}\sqrt{\frac{n}{d}}\sqrt{\frac{\pi}{2}}$

Together:

In[∘]:=
```
FullSimplify[2^(d (-1/2 + d + n)) n^(-d^2/2) π^(-d/2) BarnesG[1 + d] * Sqrt[n/d] Sqrt[π/2], {d > 0, n > 0}]
```

Out[∘]=

$$2^{-\frac{1}{2}+d\left(-\frac{1}{2}+d+n\right)}\,n^{-\frac{d^2}{2}}\sqrt{\frac{n}{d}}\,\pi^{\frac{1}{2}-\frac{d}{2}}\,\text{BarnesG}[1+d]$$

Special case d=8 (A005366)

In[∘]:=
```
% /. d → 8 // FullSimplify
```

Out[∘]=

$$\frac{1\,913\,625\times 4^{37+4n}}{n^{63/2}\,\pi^{7/2}}$$

Numerical check

In[12]:=
```
A005366 =
  Table[HypergeometricPFQ[{-7 - n, -6 - n, -5 - n, -4 - n, -3 - n, -2 - n, -1 - n, -n},
    {2, 3, 4, 5, 6, 7, 8}, 1], {n, 1, 5000}];
```

In[13]:=
```
Show[ListPlot[Table[A005366[[n]] / (1913625 × 4^(37+4 n) / (n^(63/2) π^(7/2))), {n, 1, Length[A005366]}]],
 Plot[1, {n, 1, Length[A005366]}, PlotStyle → Red]]
```

Out[13]=



Richardson extrapolation

In[15]:=
```
$MaxExtraPrecision = 1000;
funs[n_] := A005366[[n]] / (1913625 × 4^(37+4 n) / (n^(63/2) π^(7/2)));
Do[Print[N[Sum[(-1)^(m + j) * funs[j * Floor[Length[A005366] / m]] *
      j^(m - 1) / (j - 1) ! / (m - j) !, {j, 1, m}], 40]], {m, 1, 10}]
```

```
0.9512671745933798176447437548550528586796
0.9975421971519164199909680186646441716143
0.9999031071949074695365913255895850546605
0.9999968738773509022825684971953763396402
0.9999999137971541874424145308210383148967
0.9999999979066715426136631144513609841882
0.9999999999544718665768767816451460881308
0.9999999999990998439182542920333850909179
0.9999999999999833737681749498729423262078
0.999999999999999718579374847938052674599
```